
01 Définition

EN UNE PHRASE

DAX (Data Analysis Expressions) est le langage de formules de Power BI, de Power Pivot et d'Analy‐

sis Services. Il sert à créer des mesures, des colonnes calculées et des tables, qui calculent sur un

modèle de données plutôt que sur des cellules.

02 Où écrire du DAX ?

Objet Où le créer Quand il est calculé

Mesure Accueil > Nouvelle mesure À chaque affichage, pour chaque case d'un visuel

Colonne calculée Outils de table > Nouvelle colonne Une fois, à l'actualisation, ligne par ligne

Table calculée Modélisation > Nouvelle table Une fois, à l'actualisation

La formule s'écrit dans la barre de formule, sous le ruban, sous la forme Nom = expression.

03 Le simulateur

Clique sur les cases de la matrice : les lignes de la table qui participent au calcul s'allument. C'est le principe de

toutes les fiches de ce cours.

FICHE DE COURS GRATUITE

DAX | DÉCOUVRIR

DAX, c'est quoi ?

Power Query prépare les données, DAX les fait parler.

Chiffre d'affaires, évolution sur un an, part du total,

classement des clients : tout ce qu'un tableau de bord

affiche passe par une formule DAX.

CA

CA = SUM(Ventes[Montant])

MATRICE : CA

Catégorie CA

Informatique 2 452 €

Mobilier 4 391 €

Papeterie 330 €

Total 7 173 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/dax-cest-quoi.html

04 DAX et les formules Excel

Excel

Une formule par cellule, avec des références comme

B2:B50. Il faut la recopier et l'adapter quand le tableau

change.

DAX

Une formule par mesure, avec des noms de colonnes

comme Ventes[Montant]. Elle s'adapte toute seule à

chaque case du visuel, selon ses filtres.

Beaucoup de fonctions ont un nom proche d'Excel (SUM, AVERAGE, IF, DATE), mais la logique est différente : DAX

ne connaît pas de cellules, il connaît des tables, des colonnes et des filtres.

05 Pièges à éviter

⚠️ PIÈGE

Raisonner en cellules

Il n'y a pas de « ligne 2 » ni de « cellule du dessus » en DAX. Une mesure calcule sur toutes les lignes visibles dans

le contexte. Penser « quelles lignes sont filtrées ? » plutôt que « quelle cellule ? » débloque la plupart des

situations.

⚠️ PIÈGE

Faire en DAX ce que Power Query fait mieux

Nettoyer du texte, fractionner une colonne ou dépivoter des mois : c'est le travail de Power Query. DAX sert à cal‐

culer, pas à préparer.

01 Définition

EN UNE PHRASE

Une colonne calculée ajoute une colonne à une table : elle est calculée ligne par ligne à l'actualisa‐

tion et stockée. Une mesure n'est pas stockée : elle est calculée au moment de l'affichage, pour

chaque case d'un visuel, à partir des lignes visibles.

02 Le simulateur

Compare les deux méthodes : le même CA, une fois avec une colonne stockée (en jaune dans la table), une fois

avec une mesure.

FICHE DE COURS GRATUITE

DAX | DÉCOUVRIR

Mesure ou colonne
calculée ?

C'est la première question de tous ceux qui débutent en

DAX, et la source de nombreux modèles trop lourds. La

règle tient en une phrase : si on veut l'afficher ou le filtrer

par ligne, colonne ; si on veut l'agréger, mesure.

jour

COLONNE CALCULÉE

Montant = Ventes[Qté] * Ventes[Prix]

MATRICE : SOMME DE MONTANT

Catégorie Somme de Montant

Informatique 2 452 €

Mobilier 4 391 €

Papeterie 330 €

Total 7 173 €

MESURE

CA = SUMX(Ventes, Ventes[Qté] * Ventes[Prix])

MATRICE : CA

Catégorie CA

Informatique 2 452 €

Mobilier 4 391 €

Papeterie 330 €

Total 7 173 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/mesure-ou-colonne-calculee.html

03 Comment choisir

Besoin Le bon choix

Afficher une valeur par ligne, filtrer ou trier dessus Colonne calculée

Mettre en ligne ou en colonne d'une matrice, dans un segment Colonne calculée

Un total, une moyenne, un pourcentage, une évolution Mesure

Un calcul qui doit suivre les segments et les filtres Mesure

💡 ASTUCE

La règle du doute

Dans le doute, choisis la mesure. Elle ne pèse rien dans le fichier, elle suit tous les filtres et elle peut toujours être

transformée en colonne plus tard si besoin.

04 Pièges à éviter

⚠️ PIÈGE

Une colonne pour un pourcentage

Une colonne « % du total » calculée ligne par ligne ne s'additionne pas et ne réagit pas aux segments. Un pourcen‐

tage est presque toujours une mesure.

⚠️ PIÈGE

Des colonnes calculées partout

Chaque colonne calculée est stockée et alourdit le fichier, surtout sur des millions de lignes. Si la colonne peut être

créée dans Power Query, c'est encore mieux : elle est compressée plus efficacement.

01 Définition

EN UNE PHRASE

Une formule DAX s'écrit Nom = expression. Les colonnes se citent avec leur table, Ventes[Montant],

les mesures sans table, [CA]. VAR stocke un résultat intermédiaire et RETURN renvoie le résultat

final.

02 Les références

Écriture Ce que c'est

Ventes[Montant] La colonne Montant de la table Ventes

'Ventes 2026'[Montant] Apostrophes obligatoires si le nom de table contient un espace

[CA] Une mesure, toujours sans nom de table

"Nord" Un texte, entre guillemets

5% ou 0.05 Un nombre : le séparateur décimal est le point

03 Les opérateurs

Opérateur Rôle

+ - * / Calculs

& Assembler du texte : "CA de " & [Région]

= <> > < >= <= Comparaisons

&& et la double barre verticale ET et OU logiques (ou les fonctions AND et OR)

FICHE DE COURS GRATUITE

DAX | DÉCOUVRIR

La syntaxe DAX

Une formule DAX bien écrite se lit comme une phrase. Trois

conventions suffisent : la table devant chaque colonne,

jamais devant une mesure, et des variables pour découper

le calcul.

Opérateur Rôle

IN { } Appartenance : Produits[Catégorie] IN {"Mobilier", "Papeterie"}

04 Le simulateur : VAR et RETURN

PANIER MOYEN AVEC VAR

Panier moyen =
VAR ca = SUM(Ventes[Montant])
VAR nb = COUNTROWS(Ventes)
RETURN
    DIVIDE(ca, nb)

MATRICE : PANIER MOYEN

Région Panier moyen

Nord 222,08 €

Sud 209,62 €

Est 356,60 €

Total 239,10 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/syntaxe-dax-var.html

MISE EN FORME LISIBLE

Panier moyen =
VAR ca = SUM(Ventes[Montant])        // chiffre d'affaires
VAR nb = COUNTROWS(Ventes)           // nombre de ventes
RETURN
    DIVIDE(ca, nb)

💡 ASTUCE

Mets en forme tes formules

Maj + Entrée passe à la ligne dans la barre de formule. Le site gratuit DAX Formatter remet en forme n'importe

quelle formule en un clic.

05 Pièges à éviter

⚠️ PIÈGE

Le point-virgule d'Excel

En DAX dans Power BI, les arguments sont séparés par des virgules et les décimales par un point, comme en an‐

glais. ROUND([CA]; 2)  provoque une erreur.

⚠️ PIÈGE

Une table devant une mesure

Ventes[CA]  pour une mesure fonctionne, mais trompe le lecteur, qui croit lire une colonne. La convention : co‐

lonne avec sa table, mesure sans.

01 Définition

EN UNE PHRASE

Le modèle en étoile place une table de faits (les ventes, une ligne par événement) au centre, reliée à

des tables de dimensions (Produits, Clients, Dates) par des relations un à plusieurs. Les filtres posés

sur une dimension descendent vers la table de faits.

02 Le modèle « Boutique »

Table Rôle Clé

Ventes Faits : une ligne par vente, avec la quantité Réf, Code client, Date

Produits Dimension : produit, catégorie, prix Réf

Clients Dimension : client, région, type Code client

Dates Dimension : une ligne par jour, mois, année Date

03 Le simulateur : une relation qui manque

La matrice affiche la région, colonne de la table Clients. Supprime la relation pour voir ce qui se passe.

FICHE DE COURS GRATUITE

DAX | DÉCOUVRIR

Le modèle en étoile
et les relations
Une mesure DAX n'est jamais meilleure que le modèle sur

lequel elle calcule. Une table de faits au centre, des tables

de dimensions autour, des relations qui filtrent dans le bon

sens : c'est le socle de tout le reste.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau 1 / 3, Découverte

Durée de lecture 8 min

Auteur Clément Delobel

Version en ligne

RELATION ACTIVE

CA = SUM(Ventes[Montant])

MATRICE : CA

Région CA

Nord 2 665 €

Sud 2 725 €

Est 1 783 €

Ouest

Total 7 173 €

RELATION SUPPRIMÉE

CA = SUM(Ventes[Montant])

MATRICE : CA

Région CA

Nord 7 173 €

Sud 7 173 €

Est 7 173 €

Ouest 7 173 €

Total 7 173 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/modele-etoile-relations.html

04 Pièges à éviter

⚠️ PIÈGE

La relation dans le mauvais sens

Le filtre va du côté « 1 » (la dimension) vers le côté « plusieurs » (les faits). Une matrice qui met en ligne une co‐

lonne de Ventes et calcule une mesure sur Clients affichera le même chiffre partout.

⚠️ PIÈGE

Les relations bidirectionnelles

Cocher « Les deux » dans le sens du filtre croisé semble régler tous les problèmes, mais crée des ambiguïtés et

des lenteurs. Garde le sens unique par défaut, et utilise CROSSFILTER dans une mesure si besoin.

⚠️ PIÈGE

Tout dans une seule table

Une grande table plate qui répète le nom du client et la catégorie sur chaque vente marche au début, puis com‐

plique toutes les mesures. Sépare faits et dimensions, de préférence dès Power Query.

01 Définition

EN UNE PHRASE

SUM, AVERAGE, MIN et MAX agrègent une colonne sur les lignes visibles dans le contexte : somme,

moyenne, plus petite et plus grande valeur. Elles ignorent les valeurs vides.

02 Le simulateur

Change de fonction avec les boutons, puis compare les lignes et le total.

FICHE DE COURS GRATUITE

DAX | AGRÉGER

SUM, AVERAGE,
MIN, MAX

SUM, AVERAGE, MIN et MAX ressemblent à leurs cousines

d'Excel. Une différence de taille : elles portent sur une

colonne entière, filtrée par le contexte de la case, jamais sur

une plage de cellules.

SUM

CA = SUM(Ventes[Montant])

MATRICE : CA

Catégorie CA

Informatique 2 452 €

Mobilier 4 391 €

Papeterie 330 €

Total 7 173 €

AVERAGE

Montant moyen = AVERAGE(Ventes[Montant])

MATRICE : MONTANT MOYEN

Catégorie Montant moyen

Informatique 188,62 €

Mobilier 337,77 €

Papeterie 82,50 €

Total 239,10 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/sum-average-min-max.html

LES QUATRE MESURES

CA = SUM(Ventes[Montant])
Montant moyen = AVERAGE(Ventes[Montant])
Plus petite vente = MIN(Ventes[Montant])
Plus grosse vente = MAX(Ventes[Montant])

💡 ASTUCE

Une mesure explicite plutôt qu'un glisser-déposer

Glisser la colonne Montant dans un visuel crée une somme implicite. Écrire la mesure CA permet de la réutiliser

dans d'autres mesures, de la nommer et de la mettre en forme une fois pour toutes.

03 Pièges à éviter

⚠️ PIÈGE

Le total d'une moyenne

Sur la ligne Total, AVERAGE fait la moyenne de toutes les lignes visibles, pas la moyenne des moyennes de chaque

catégorie. Les deux résultats diffèrent dès que les catégories n'ont pas le même nombre de ventes.

⚠️ PIÈGE

Une colonne de texte

SUM sur une colonne de type texte renvoie une erreur. Vérifie le type de la colonne dans Power Query ou dans la

vue Données.

01 Définition

EN UNE PHRASE

COUNTROWS compte les lignes d'une table, COUNT les valeurs non vides d'une colonne, DIS‐

TINCTCOUNT les valeurs différentes d'une colonne. Toutes travaillent sur les lignes visibles dans le

contexte.

02 Le simulateur

Passe sur DISTINCTCOUNT et regarde la ligne Total.

FICHE DE COURS GRATUITE

DAX | AGRÉGER

Compter :
COUNTROWS,
COUNT,
DISTINCTCOUNT

« Combien de ventes ? » et « combien de clients ? » ne sont

pas la même question. COUNTROWS répond à la première,

DISTINCTCOUNT à la seconde, et la ligne Total le montre

bien.

Niveau 1 / 3, Découverte

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
compter-countrows-dist
inctcount.html

COUNTROWS

Nb ventes = COUNTROWS(Ventes)

MATRICE : NB VENTES

Catégorie Nb ventes

Informatique 13

Mobilier 13

Papeterie 4

Total 30

COUNT

Nb ventes = COUNT(Ventes[Qté])

MATRICE : NB VENTES

Catégorie Nb ventes

Informatique 13

Mobilier 13

Papeterie 4

Total 30

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/compter-countrows-distinctcount.html

03 Les fonctions de comptage

Fonction Compte Exemple

COUNTROWS(Ventes) Les lignes de la table Nombre de ventes

COUNT(Ventes[Qté]) Les nombres non vides d'une colonne Ventes avec une quantité

COUNTA(Ventes[Réf]) Les valeurs non vides, tout type Ventes avec une référence

DISTINCTCOUNT(Ventes[Code client]) Les valeurs différentes Nombre de clients acheteurs

COUNTBLANK(Ventes[Remise]) Les valeurs vides Ventes sans remise

04 À toi de jouer

DÉFI DAX N°1 | 10 MIN

Ventes et clients par région

Le directeur commercial veut savoir, pour 2026, combien de ventes ont été faites dans chaque région, et par

combien de clients différents.

Ce que tu dois obtenir

Région Nb ventes Nb clients

Nord 84 3

Sud 70 3

Est 38 2

Ouest 36 2

Résultat attendu : Une matrice Région en lignes, deux mesures, un segment Année réglé sur 2026.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/compter-countrows-distinctcount.html

05 Pièges à éviter

⚠️ PIÈGE

Additionner des clients distincts

Le total de DISTINCTCOUNT n'est pas la somme des lignes : un client qui achète dans deux catégories compte une

fois au total. C'est juste, mais surprenant. Ne recalcule jamais ce total à la main.

⚠️ PIÈGE

Compter sur la mauvaise table

COUNTROWS(Clients) compte les clients du fichier, pas ceux qui ont acheté. Pour les acheteurs, compte les codes

clients dans la table Ventes.

01 Définition

EN UNE PHRASE

DIVIDE(numérateur, dénominateur, [remplacement]) divise le premier argument par le second et

renvoie la valeur de remplacement (vide par défaut) si le dénominateur vaut zéro ou vide.

02 Le simulateur

La Papeterie n'existait pas en 2025 : compare l'opérateur / et DIVIDE.

FICHE DE COURS GRATUITE

DAX | AGRÉGER

DIVIDE et les
pourcentages
Une évolution sur un an, une marge en pourcentage, un taux

de transformation : tout pourcentage est une division, et

toute division finit un jour par un dénominateur vide. DIVIDE

l'anticipe.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau 1 / 3, Découverte

Durée de lecture 6 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
divide.html

AVEC L'OPÉRATEUR /

Évolution = ([CA 2026] - [CA 2025]) / [CA 2025]

MATRICE : ÉVOLUTION

Catégorie Évolution

Informatique 36,5 %

Mobilier 39,7 %

Papeterie Infinity

Total 50,0 %

AVEC DIVIDE

Évolution = DIVIDE([CA 2026] - [CA 2025], [CA 2025])

MATRICE : ÉVOLUTION

Catégorie Évolution

Informatique 36,5 %

Mobilier 39,7 %

Papeterie

Total 50,0 %

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/divide.html

UNE ÉVOLUTION, ET UNE VALEUR DE REMPLACEMENT

Évolution = DIVIDE([CA 2026] - [CA 2025], [CA 2025])
Évolution (texte) = IF(ISBLANK([CA 2025]), "Nouveau", FORMAT([Évolution], "0.0 %"))

💡 ASTUCE

Mets la mesure au format pourcentage

Plutôt que de multiplier par 100, laisse le résultat entre 0 et 1 et choisis le format Pourcentage dans Outils de me‐

sure. Le tri, les totaux et les graphiques restent justes.

03 Pièges à éviter

⚠️ PIÈGE

Infinity dans un visuel

L'opérateur / renvoie Infinity quand on divise par zéro ou par vide, et NaN pour zéro divisé par zéro. Le visuel af‐

fiche ces valeurs telles quelles, et les graphiques deviennent illisibles.

⚠️ PIÈGE

DIVIDE partout

DIVIDE est un peu plus lent que / sur de très gros volumes. Si le dénominateur ne peut jamais être nul (une

constante, par exemple), l'opérateur / suffit.

01 Définition

EN UNE PHRASE

Un itérateur parcourt une table ligne par ligne, évalue une expression pour chaque ligne, puis agrège

les résultats. SUMX(table, expression) additionne, AVERAGEX fait la moyenne, MINX et MAXX

gardent le minimum et le maximum.

02 Le simulateur

Clique sur une catégorie, puis sur « Voir le calcul ligne par ligne ». Compare ensuite avec SUM × SUM.

FICHE DE COURS GRATUITE

DAX | AGRÉGER

Les itérateurs :
SUMX, AVERAGEX

Quand le montant n'existe pas dans la table, il faut le

calculer ligne par ligne avant de l'additionner. C'est

exactement ce que fait SUMX, et le simulateur le déroule

sous tes yeux.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 9 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
iterateurs-sumx.html

SUMX (JUSTE)

CA = SUMX(Ventes, Ventes[Qté] * Ventes[Prix])

MATRICE : CA

Catégorie CA

Informatique 2 452 €

Mobilier 4 391 €

Papeterie 330 €

Total 7 173 €

SUM × SUM (FAUX)

CA faux = SUM(Ventes[Qté]) * SUM(Ventes[Prix])

MATRICE : CA FAUX

Catégorie CA faux

Informatique 44 676 €

Mobilier 68 042 €

Papeterie 1 680 €

Total 397 800 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/iterateurs-sumx.html

03 Aller chercher le prix dans une autre table

Dans le vrai modèle, le prix est dans la table Produits. RELATED permet de le lire depuis chaque ligne de Ventes,

grâce à la relation.

CA AVEC LE PRIX DE LA TABLE PRODUITS

CA = SUMX(Ventes, Ventes[Qté] * RELATED(Produits[Prix]))

Itérateur Résultat

SUMX Somme des résultats ligne par ligne

AVERAGEX Moyenne des résultats

MINX, MAXX Plus petit, plus grand résultat

COUNTX Nombre de résultats non vides

RANKX Rang, voir la fiche RANKX et TOPN

04 À toi de jouer

DÉFI DAX N°2 | 10 MIN

Le CA sans colonne Montant

La table Ventes ne contient que la quantité. Le prix est dans la table Produits. Calcule le CA par catégorie

avec une seule mesure, sans créer de colonne calculée.

Ce que tu dois obtenir

Catégorie CA

Informatique 38 228 €

Mobilier 45 332 €

Papeterie 26 566 €

Total 110 126 €

Résultat attendu : Une mesure CA, une matrice par catégorie, les deux années confondues.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/iterateurs-sumx.html

05 Pièges à éviter

⚠️ PIÈGE

SUM(Qté) * SUM(Prix)

Multiplier deux totaux n'est pas additionner des produits. Dès qu'il y a plus d'une ligne, le résultat est faux, et sou‐

vent très faux.

⚠️ PIÈGE

Itérer sur une table énorme

SUMX sur une table de 50 millions de lignes avec une expression complexe peut être lent. Garde l'expression

simple, et précalcule dans Power Query ce qui peut l'être.

01 Définition

EN UNE PHRASE

Le contexte de filtre est l'ensemble des filtres qui s'appliquent au moment où une mesure est calcu‐

lée : la ligne et la colonne de la matrice, les segments, les filtres du visuel, de la page et du rapport.

La mesure ne voit que les lignes qui passent tous ces filtres.

02 Le simulateur

Clique sur plusieurs cases, puis sur les totaux, et change le segment Année. Le panneau sous la matrice liste les

filtres de la case.

FICHE DE COURS GRATUITE

DAX | CONTEXTE

Le contexte de filtre

Une seule mesure, et pourtant un résultat différent dans

chaque case : c'est le contexte de filtre. Une fois qu'on le

voit, DAX devient logique. Le simulateur le montre case par

case.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 9 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
contexte-de-filtre.htm
l

SEGMENT ANNÉE = 2026

CA = SUM(Ventes[Montant])

MATRICE : CA

Catégorie Nord Sud Est Total

Informatique 714 € 603 € 98 € 1 415 €

Mobilier 610 € 874 € 1 075 € 2 559 €

Papeterie 150 € 180 € 330 €

Total 1 474 € 1 657 € 1 173 € 4 304 €

SEGMENT ANNÉE = 2025

CA = SUM(Ventes[Montant])

MATRICE : CA

Catégorie Nord Sud Est Total

Informatique 434 € 603 € 1 037 €

Mobilier 757 € 465 € 610 € 1 832 €

Papeterie

Total 1 191 € 1 068 € 610 € 2 869 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/contexte-de-filtre.html

03 D'où viennent les filtres ?

Source Exemple

Ligne et colonne du visuel Catégorie = Mobilier, Région = Sud

Segments Année = 2026

Volet Filtres (visuel, page, rapport) Type de client = Professionnel

Interactions entre visuels Clic sur une barre d'un autre graphique

CALCULATE dans la mesure Voir la fiche CALCULATE

Les filtres s'additionnent : une ligne doit les respecter tous. Et grâce aux relations, un filtre posé sur Produits ou

Clients descend jusqu'à la table Ventes.

04 Le cas des totaux

Une ligne Total n'additionne pas les cases du dessus : elle recalcule la mesure avec moins de filtres (sans le filtre

de la ligne). Pour SUM, cela revient au même. Pour une moyenne, un comptage distinct ou une condition, le total

peut surprendre : voir Mon total est faux.

05 À toi de jouer

DÉFI DAX N°3 | 8 MIN

Lire les contextes de filtre

Avec la seule mesure CA = SUMX(Ventes, Ventes[Qté] * RELATED(Produits[Prix])), une matrice Catégorie ×

Région et un segment Année, trouve les trois valeurs ci-dessous sans écrire de nouvelle mesure.

Ce que tu dois obtenir

Question Réponse

Le CA Mobilier dans le Sud en 2025 6 024 €

Le CA Informatique tous lieux confondus en 2026 23 525 €

Le CA de la région Ouest sur les deux années 16 829 €

Résultat attendu : Trois valeurs lues dans la même matrice, en jouant sur le segment et les totaux.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/contexte-de-filtre.html

06 Pièges à éviter

⚠️ PIÈGE

Chercher à écrire une mesure par case

Pas besoin d'une mesure « CA Mobilier Sud 2025 » : la mesure CA, placée dans la bonne matrice avec le bon seg‐

ment, la calcule déjà. On n'écrit une mesure spécifique que pour modifier le contexte, avec CALCULATE.

⚠️ PIÈGE

Oublier un filtre de page

Un chiffre « faux » vient souvent d'un filtre oublié dans le volet Filtres, au niveau de la page ou du rapport. Vérifie-

le avant de soupçonner la mesure.

mesétudesdecas.fr
DAX | Recueil complet

V1.0 - 02/11/2026

FALE | Formation, Accompagnement, Logiciels, Expertise | EURL au capital de 2 000 € | RCS Lille Métropole n° 933 672 537
229 rue de Solférino, 59000 Lille | TVA FR33 933 672 537 | c.delobel@fale.fr | Toutes les fiches gratuites : mesetudesdecas.fr/dax | © FALE

Page 29 / 94



01 Définition

EN UNE PHRASE

Le contexte de ligne existe quand DAX parcourt une table ligne par ligne : dans une colonne calcu‐

lée ou dans un itérateur (SUMX, FILTER…). Une référence de colonne y renvoie la valeur de la ligne

en cours. Il ne filtre pas les autres tables.

02 Le simulateur

La colonne Taille est calculée ligne par ligne. Lance le calcul pour voir le contexte de ligne en action.

COLONNE CALCULÉE

Taille = IF(Ventes[Montant] >= 300, "Grosse", "Petite")

MATRICE : NB GROSSES VENTES

Catégorie Nb grosses ventes

Informatique 3

Mobilier 7

Papeterie 0

Total 10

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/contexte-de-ligne.html

FICHE DE COURS GRATUITE

DAX | CONTEXTE

Le contexte de ligne

Dans une colonne calculée, DAX sait sur quelle ligne il se

trouve : Ventes[Montant] y vaut le montant de cette ligne.

C'est le contexte de ligne. Il est simple, mais il ne filtre rien,

et c'est là que tout se joue.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
contexte-de-ligne.html

03 Deux contextes à ne pas confondre

Contexte de ligne

« Je suis sur la vente V07. » Une colonne renvoie la va‐

leur de cette ligne. Il existe dans les colonnes calculées

et les itérateurs.

Contexte de filtre

« Je ne vois que les ventes de Mobilier en 2026. » Il ré‐

duit les lignes visibles. Il existe dans les mesures et

vient du visuel ou de CALCULATE.

COMPARER À LA MOYENNE, LIGNE PAR LIGNE

Au-dessus de la moyenne =
VAR montant = Ventes[Montant]
RETURN
    IF(montant > AVERAGE(Ventes[Montant]), "Oui", "Non")

💡 ASTUCE

Les variables remplacent EARLIER

L'ancienne fonction EARLIER sert à retrouver la ligne en cours dans un calcul imbriqué. Une variable, comme mon‐

tant ci-dessus, fait la même chose de façon bien plus lisible.

04 Pièges à éviter

⚠️ PIÈGE

Une colonne dans une mesure, sans agrégation

CA = Ventes[Montant]  dans une mesure renvoie une erreur : une mesure n'a pas de ligne en cours. Il faut agré‐

ger (SUM) ou itérer (SUMX).

⚠️ PIÈGE

Croire que la ligne filtre la table liée

Dans une colonne calculée de Clients, SUM(Ventes[Montant]) renvoie le total de toutes les ventes. Pour filtrer par

le client de la ligne, il faut CALCULATE : voir La transition de contexte.

01 Définition

EN UNE PHRASE

CALCULATE(expression, filtre1, filtre2, …) évalue l'expression dans un contexte de filtre modifié :

chaque filtre remplace le filtre existant sur la même colonne, ou s'ajoute s'il porte sur une autre

colonne.

02 Le simulateur

Passe sur « CALCULATE (remplace) » et clique sur Mobilier : le filtre de la ligne est barré, remplacé par Informa‐

tique. Compare avec KEEPFILTERS.

FICHE DE COURS GRATUITE

DAX | CONTEXTE

CALCULATE

CALCULATE est la fonction qui distingue DAX d'Excel. Elle

calcule une expression dans un contexte de filtre modifié :

on lui dit « calcule le CA, mais seulement pour l'Informatique

», et elle s'occupe du reste.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 10 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
calculate.html

CA

CA = SUM(Ventes[Montant])

MATRICE : CA

Catégorie CA

Informatique 2 452 €

Mobilier 4 391 €

Papeterie 330 €

Total 7 173 €

CALCULATE (REMPLACE)

CA Informatique = CALCULATE([CA], Produits[Catégorie] = "Informatique")

MATRICE : CA INFORMATIQUE

Catégorie CA Informatique

Informatique 2 452 €

Mobilier 2 452 €

Papeterie 2 452 €

Total 2 452 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/calculate.html

03 Exemples

UN FILTRE, PLUSIEURS FILTRES, UNE LISTE

CA Informatique = CALCULATE([CA], Produits[Catégorie] = "Informatique")

CA Pro Nord = CALCULATE([CA], Clients[Type] = "Professionnel", Clients[Région] = "Nord")

CA Bureau = CALCULATE([CA], Produits[Catégorie] IN {"Mobilier", "Papeterie"})

Ce que fait le filtre Effet sur le contexte

Porte sur une colonne déjà filtrée Remplace le filtre existant

Porte sur une autre colonne S'ajoute aux filtres existants

Est enveloppé dans KEEPFILTERS S'ajoute toujours, même sur la même colonne

Est un ALL ou un REMOVEFILTERS Retire des filtres : voir la part du total

04 À toi de jouer

DÉFI DAX N°4 | 12 MIN

Le poids des professionnels

Pour chaque région, calcule le CA réalisé avec les clients professionnels (colonne Clients[Type]) et sa part

dans le CA de la région.

Ce que tu dois obtenir

Région CA CA Pro % Pro

Nord 38 040 € 34 500 € 90,7 %

Sud 37 816 € 28 910 € 76,4 %

Est 17 441 € 14 133 € 81,0 %

Ouest 16 829 € 8 529 € 50,7 %

Résultat attendu : Trois mesures, deux années confondues.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/calculate.html

05 Pièges à éviter

⚠️ PIÈGE

Le filtre qui écrase la ligne

Dans une matrice par catégorie, CA Informatique affiche la même valeur sur toutes les lignes. Ce n'est pas un bug :

le filtre de la ligne est remplacé. Si tu veux respecter la ligne, utilise KEEPFILTERS.

⚠️ PIÈGE

Une mesure dans la condition

CALCULATE([CA], [Marge] > 100)  est refusé : une condition simple porte sur une colonne. Pour filtrer sur une

mesure, il faut FILTER sur une table : voir FILTER.

01 Définition

EN UNE PHRASE

Utilisées dans CALCULATE, ALL(table ou colonne) et REMOVEFILTERS retirent les filtres d'une table

ou d'une colonne, ALLEXCEPT(table, colonne) les retire tous sauf ceux de la colonne indiquée. On

s'en sert pour calculer un dénominateur de pourcentage.

02 Le simulateur

Compare les deux dénominateurs : la somme de toute la matrice fait 100 %, ou chaque colonne fait 100 %.

FICHE DE COURS GRATUITE

DAX | CONTEXTE

ALL, ALLEXCEPT,
REMOVEFILTERS : la
part du total

« Quelle part du CA fait chaque catégorie ? » La réponse est

une division, mais le dénominateur doit ignorer certains

filtres. ALL et ses cousines retirent exactement ceux qu'on

leur indique.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 9 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
all-part-du-total.html

% DU TOTAL GÉNÉRAL

% du total = DIVIDE([CA], CALCULATE([CA], ALL(Ventes)))

MATRICE : % DU TOTAL

Catégorie Nord Sud Est Total

Informatique 16,0 % 16,8 % 1,4 % 34,2 %

Mobilier 19,1 % 18,7 % 23,5 % 61,2 %

Papeterie 2,1 % 2,5 % 4,6 %

Total 37,2 % 38,0 % 24,9 % 100,0 %

% DANS LA RÉGION

% dans la région = DIVIDE([CA], CALCULATE([CA], ALL(Produits[Catégorie])))

MATRICE : % DANS LA RÉGION

Catégorie Nord Sud Est Total

Informatique 43,1 % 44,3 % 5,5 % 34,2 %

Mobilier 51,3 % 49,1 % 94,5 % 61,2 %

Papeterie 5,6 % 6,6 % 4,6 %

Total 100,0 % 100,0 % 100,0 % 100,0 %

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/all-part-du-total.html

03 Quel filtre retirer ?

Dénominateur voulu Écriture

Total général CALCULATE([CA], ALL(Ventes)) ou REMOVEFILTERS()

Total de toutes les catégories, région gardée CALCULATE([CA], ALL(Produits[Catégorie]))

Total de la catégorie (pour chaque produit) CALCULATE([CA], ALLEXCEPT(Produits, Produits[Catégorie]))

Total de ce qui est sélectionné dans les segments CALCULATE([CA], ALLSELECTED(Produits))

PART DU TOTAL GÉNÉRAL

% du total = DIVIDE([CA], CALCULATE([CA], ALL(Ventes)))

04 À toi de jouer

DÉFI DAX N°5 | 12 MIN

Le poids de chaque produit dans sa catégorie

Dans une matrice Catégorie puis Produit, affiche pour chaque produit sa part dans le CA de sa catégorie.

Chaque catégorie doit faire 100 %.

Ce que tu dois obtenir

Produit (Informatique) CA % de la catégorie

Clavier 8 428 € 22,0 %

Souris 4 325 € 11,3 %

Écran 27 pouces 13 230 € 34,6 %

Casque 12 245 € 32,0 %

Informatique 38 228 € 100,0 %

Résultat attendu : Même logique pour Mobilier et Papeterie.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/all-part-du-total.html

05 Pièges à éviter

⚠️ PIÈGE

ALL qui ignore les segments

ALL(Ventes) retire aussi le filtre du segment Année : le pourcentage est calculé sur les deux années. Pour respec‐

ter les segments mais pas les lignes du visuel, utilise ALLSELECTED.

⚠️ PIÈGE

ALL sur la mauvaise colonne

ALL(Ventes[Catégorie]) ne sert à rien si la matrice filtre par Produits[Catégorie] : il faut retirer le filtre sur la colonne

réellement utilisée dans le visuel.

01 Définition

EN UNE PHRASE

FILTER(table, condition) parcourt la table ligne par ligne et renvoie les lignes pour lesquelles la

condition est vraie. C'est un itérateur : la condition est évaluée dans un contexte de ligne.

02 Le simulateur

CA DES GROSSES VENTES

CA grosses ventes =
CALCULATE(
    [CA],
    FILTER(Ventes, Ventes[Montant] >= 300)
)

MATRICE : CA GROSSES VENTES

Catégorie CA grosses ventes

Informatique 1 134 €

Mobilier 3 315 €

Papeterie

Total 4 449 €

FICHE DE COURS GRATUITE

DAX | CONTEXTE

FILTER

FILTER renvoie une table : les lignes qui remplissent une

condition. Dans CALCULATE, elle permet des filtres

impossibles autrement, comme « les ventes de plus de 300

€ » ou « les clients dont le CA dépasse 1 000 € ».

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
filter.html

mesétudesdecas.fr
DAX | Recueil complet

V1.0 - 02/11/2026

FALE | Formation, Accompagnement, Logiciels, Expertise | EURL au capital de 2 000 € | RCS Lille Métropole n° 933 672 537
229 rue de Solférino, 59000 Lille | TVA FR33 933 672 537 | c.delobel@fale.fr | Toutes les fiches gratuites : mesetudesdecas.fr/dax | © FALE

Page 38 / 94



Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/filter.html

03 Condition simple ou FILTER ?

LES DEUX ÉCRITURES

CA grosses ventes = CALCULATE([CA], Ventes[Montant] >= 300)
CA grosses ventes = CALCULATE([CA], FILTER(Ventes, Ventes[Montant] >= 300))

CA bons clients = CALCULATE([CA], FILTER(VALUES(Clients[Client]), [CA] > 1000))

Une condition sur une colonne : la forme simple suffit, elle est plus rapide.

Une condition sur une mesure ou sur plusieurs colonnes de tables différentes : FILTER est nécessaire.

Filtre une petite table (VALUES d'une colonne) plutôt que toute la table de faits.

04 Pièges à éviter

⚠️ PIÈGE

FILTER(Ventes, …) sur des millions de lignes

Parcourir toute la table de faits est lent. Préfère FILTER(VALUES(Clients[Client]), …) : quelques centaines de clients

plutôt que des millions de ventes.

⚠️ PIÈGE

FILTER(ALL(...)) sans le vouloir

FILTER(ALL(Produits), …) ignore la catégorie de la ligne du visuel. C'est utile pour un classement, mais surprenant

ailleurs : vérifie que tu veux bien retirer ces filtres.

01 Définition

EN UNE PHRASE

La transition de contexte est l'effet de CALCULATE quand il est évalué dans un contexte de ligne : la

ligne en cours devient un filtre sur chacune de ses colonnes. Toute mesure appelée dans un

contexte de ligne déclenche la même transition, car une mesure contient un CALCULATE implicite.

02 Le simulateur

Ici la table affichée est Clients. Compare la colonne calculée avec et sans CALCULATE.

FICHE DE COURS GRATUITE

DAX | CONTEXTE

La transition de
contexte
Dans une colonne calculée de la table Clients,

SUM(Ventes[Montant]) renvoie le total de la boutique sur

chaque ligne. Ajoute CALCULATE autour, et chaque client

affiche enfin son propre CA. Ce tour de magie a un nom.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau 3 / 3, Avancé

Durée de lecture 9 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
transition-de-context
e.html

SUM SEUL (FAUX)

CA client = SUM(Ventes[Montant])

MATRICE : CA CLIENT

Client CA client

Garage Martin 7 173 €

Cabinet Roy 7 173 €

Fleurs & Co 7 173 €

Café des Arts 7 173 €

Librairie Page 7 173 €

Atelier Lumière 7 173 €

AVEC CALCULATE (JUSTE)

CA client = CALCULATE(SUM(Ventes[Montant]))

MATRICE : CA CLIENT

Client CA client

Garage Martin 1 541 €

Cabinet Roy 1 736 €

Fleurs & Co 1 124 €

Café des Arts 1 783 €

Librairie Page 989 €

Atelier Lumière

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/transition-de-contexte.html

03 Où elle se produit

Situation Transition ?

Colonne calculée avec CALCULATE ou une mesure Oui

Itérateur (SUMX, FILTER…) qui appelle une mesure Oui, à chaque ligne

Mesure placée dans un visuel Non : il n'y a pas de contexte de ligne

LES DEUX ÉCRITURES ÉQUIVALENTES DANS UNE COLONNE DE CLIENTS

CA client = CALCULATE(SUM(Ventes[Montant]))
CA client = [CA]

04 Pièges à éviter

⚠️ PIÈGE

Une transition sur une table sans clé

La ligne en cours devient un filtre sur toutes ses colonnes. Si deux lignes de la table sont identiques, elles se filtrent

mutuellement et le résultat est compté deux fois. Itère sur une table qui a une clé unique.

⚠️ PIÈGE

Une mesure dans SUMX sur la table de faits

SUMX(Ventes, [CA]) déclenche une transition pour chaque vente : c'est lent, et rarement ce qu'on veut. Itère plutôt

sur une dimension : SUMX(VALUES(Clients[Client]), [CA]).

01 Définition

EN UNE PHRASE

Une table de dates contient une ligne par jour, sans trou, sur toute la période des données, avec des

colonnes Année, Mois, Trimestre… Reliée à la colonne date des ventes et marquée comme table de

dates, elle permet l'intelligence temporelle.

02 Le simulateur

Compare une matrice avec le mois en texte et une matrice avec la table de dates.

FICHE DE COURS GRATUITE

DAX | TEMPS

La table de dates

Toutes les fonctions de temps de DAX (cumul, N-1, moyenne

mobile) supposent une table de dates complète, reliée et

marquée. Sans elle, les mois se trient par ordre alphabétique

et rien ne fonctionne.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 8 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
table-de-dates.html

MOIS EN TEXTE, SANS TABLE DE DATES

CA = SUM(Ventes[Montant])

MATRICE : CA 2026

Mois CA 2026

Avr 538 €

Févr 873 €

Janv 847 €

Juin 871 €

Mai 758 €

Mars 417 €

Total 4 304 €

TABLE DATES TRIÉE PAR NUMÉRO DE MOIS

Dates = CALENDAR(DATE(2025, 1, 1), DATE(2026, 12, 31))

MATRICE : CA 2026

Mois CA 2026

Janv 847 €

Févr 873 €

Mars 417 €

Avr 538 €

Mai 758 €

Juin 871 €

Total 4 304 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/table-de-dates.html

03 Créer la table en DAX

TABLE DE DATES AVEC SES COLONNES

Dates =
ADDCOLUMNS(
    CALENDAR(DATE(2025, 1, 1), DATE(2026, 12, 31)),
    "Année", YEAR([Date]),
    "Num mois", MONTH([Date]),
    "Mois", FORMAT([Date], "mmmm"),
    "Trimestre", "T" & QUARTER([Date])
)

04 Pièges à éviter

⚠️ PIÈGE

Une table de dates avec des trous

Une table créée à partir des seules dates de vente saute les jours sans vente : les fonctions de temps donnent alors

des résultats faux. CALENDAR génère tous les jours.

⚠️ PIÈGE

Des dates avec une heure

Si Ventes[Date] contient aussi l'heure (02/03/2026 14:35), la relation ne trouve aucune correspondance. Passe la

colonne en type Date dans Power Query.

Modélisation > Nouvelle table, colle la formule.1

Outils de table > Marquer comme table de dates, colonne Date.2

Sélectionne la colonne Mois, puis Trier par colonne > Num mois.3

Dans la vue Modèle, relie Dates[Date] à Ventes[Date].4

01 Définition

EN UNE PHRASE

TOTALYTD(expression, Dates[Date]) calcule l'expression du 1er janvier jusqu'à la dernière date du

contexte. C'est un raccourci de CALCULATE(expression, DATESYTD(Dates[Date])).

02 Le simulateur

Passe sur le cumul et clique sur Avr : les ventes de janvier à avril s'allument.

FICHE DE COURS GRATUITE

DAX | TEMPS

Cumuls : TOTALYTD
et DATESYTD

« Où en est-on depuis le début de l'année ? » TOTALYTD

répond en une ligne : il remplace le mois de la case par la

période qui va du 1er janvier à la fin de ce mois.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
cumul-totalytd.html

CA DU MOIS

CA = SUM(Ventes[Montant])

MATRICE : CA

Mois CA

Janv 847 €

Févr 873 €

Mars 417 €

Avr 538 €

Mai 758 €

Juin 871 €

Total 4 304 €

CUMUL DEPUIS JANVIER

CA cumulé = TOTALYTD([CA], Dates[Date])

MATRICE : CA CUMULÉ

Mois CA cumulé

Janv 847 €

Févr 1 720 €

Mars 2 137 €

Avr 2 675 €

Mai 3 433 €

Juin 4 304 €

Total 4 304 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/cumul-totalytd.html

CUMULS ANNUEL, TRIMESTRIEL, MENSUEL, ET EXERCICE DÉCALÉ

CA cumulé = TOTALYTD([CA], Dates[Date])
CA cumulé trimestre = TOTALQTD([CA], Dates[Date])
CA cumulé mois = TOTALMTD([CA], Dates[Date])
CA cumulé exercice = TOTALYTD([CA], Dates[Date], "30/06")

03 À toi de jouer

DÉFI DAX N°6 | 10 MIN

Le cumul depuis janvier

Avec la table Dates du classeur (à marquer comme table de dates et à relier à Ventes[Date]), affiche pour

chaque mois de 2026 le CA du mois et le CA cumulé depuis le 1er janvier.

Ce que tu dois obtenir

Mois 2026 CA CA cumulé

Janvier 6 179 € 6 179 €

Février 4 976 € 11 155 €

Mars 5 086 € 16 241 €

Avril 4 870 € 21 111 €

Mai 6 071 € 27 182 €

Juin 4 801 € 31 983 €

…

Résultat attendu : 12 lignes ; en décembre, le cumul est égal au CA 2026 : 60 443 €.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/cumul-totalytd.html

04 Pièges à éviter

⚠️ PIÈGE

Le cumul qui continue dans le futur

Si la table de dates va jusqu'au 31 décembre, le cumul s'affiche aussi pour les mois sans vente, avec la même va‐

leur. Masque ces mois : IF(MIN(Dates[Date]) <= MAX(Ventes[Date]), TOTALYTD(...)).

⚠️ PIÈGE

La date de fin d'exercice

Le troisième argument de TOTALYTD est la date de fin d'exercice, au format du modèle. Pour un exercice clos au

30 juin, écris "30/06" en français, "06/30" si le modèle est en anglais.

01 Définition

EN UNE PHRASE

SAMEPERIODLASTYEAR(Dates[Date]) renvoie les mêmes dates un an plus tôt. Dans CALCULATE,

elle remplace la période de la case par celle de l'année précédente.

02 Le simulateur

Clique sur Mars en CA N-1 : ce sont les ventes de mars 2025 qui s'allument. Passe ensuite sur l'évolution.

FICHE DE COURS GRATUITE

DAX | TEMPS

Comparer à N-1

Un chiffre seul ne dit rien : 12 000 € en mars, c'est bien ou

pas ? Comparé à mars de l'année précédente, il devient une

information. SAMEPERIODLASTYEAR décale la période d'un

an, le reste est une division.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 8 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
comparer-n-1.html

CA N-1

CA N-1 = CALCULATE([CA], SAMEPERIODLASTYEAR(Dates[Date]))

MATRICE : CA N-1

Mois CA N-1

Janv 418 €

Févr 479 €

Mars 217 €

Avr 698 €

Mai 292 €

Juin 765 €

Total 2 869 €

ÉVOLUTION %

Évolution % = DIVIDE([CA] - [CA N-1], [CA N-1])

MATRICE : ÉVOLUTION %

Mois Évolution %

Janv 102,6 %

Févr 82,3 %

Mars 92,2 %

Avr -22,9 %

Mai 159,6 %

Juin 13,9 %

Total 50,0 %

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/comparer-n-1.html

CA N-1 ET ÉVOLUTION

CA N-1 = CALCULATE([CA], SAMEPERIODLASTYEAR(Dates[Date]))
Évolution % = DIVIDE([CA] - [CA N-1], [CA N-1])

Fonction Décalage

SAMEPERIODLASTYEAR(Dates[Date]) Un an en arrière, même période

DATEADD(Dates[Date], -1, MONTH) N'importe quel décalage : mois, trimestre, année

PARALLELPERIOD(Dates[Date], -1, YEAR) L'année précédente entière, quelle que soit la période

03 À toi de jouer

DÉFI DAX N°7 | 12 MIN

2026 contre 2025

Pour chaque catégorie, compare le CA 2026 au CA 2025 : affiche le CA, le CA N-1 et l'évolution en %. Le

segment Année est réglé sur 2026.

Ce que tu dois obtenir

Catégorie CA 2026 CA N-1 Évolution

Informatique 23 525 € 14 703 € 60,0 %

Mobilier 24 751 € 20 581 € 20,3 %

Papeterie 12 167 € 14 399 € -15,5 %

Résultat attendu : Trois mesures, une matrice par catégorie, segment sur 2026.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/comparer-n-1.html

04 Pièges à éviter

⚠️ PIÈGE

Le segment sur la mauvaise colonne

Si le segment Année porte sur une colonne de la table Ventes, SAMEPERIODLASTYEAR ne peut pas remonter le

temps : le filtre de l'année reste. Le segment doit porter sur la table Dates.

⚠️ PIÈGE

L'année en cours incomplète

Comparer l'année 2026 entière, arrêtée en juin, à l'année 2025 complète donne une baisse trompeuse. Compare

des périodes de même longueur, ou des cumuls à date.

01 Définition

EN UNE PHRASE

DATESINPERIOD(Dates[Date], date de départ, nombre, intervalle) renvoie une période glissante, par

exemple les 3 derniers mois jusqu'à la date de départ. Dans CALCULATE, elle remplace la période de

la case.

02 Le simulateur

Clique sur Avr : les ventes de février à avril s'allument.

MOYENNE MOBILE 3 MOIS

Moyenne 3 mois =
CALCULATE(
    [CA] / 3,
    DATESINPERIOD(Dates[Date], MAX(Dates[Date]), -3, MONTH)
)

MATRICE : MOYENNE 3 MOIS

Mois Moyenne 3 mois

Janv 282,33 €

Févr 573,33 €

Mars 712,33 €

Avr 609,33 €

Mai 571 €

Juin 722,33 €

FICHE DE COURS GRATUITE

DAX | TEMPS

La moyenne mobile

Les ventes mensuelles montent et descendent ; la tendance,

elle, se lit mieux sur une moyenne des trois derniers mois.

DATESINPERIOD fournit la fenêtre glissante, CALCULATE fait

le reste.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau 3 / 3, Avancé

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
moyenne-mobile.html

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/moyenne-mobile.html

MOYENNE DES 3 DERNIERS MOIS, MOIS PAR MOIS

Moyenne 3 mois =
CALCULATE(
    AVERAGEX(VALUES(Dates[Mois]), [CA]),
    DATESINPERIOD(Dates[Date], MAX(Dates[Date]), -3, MONTH)
)

💡 ASTUCE

AVERAGEX plutôt qu'une division par 3

AVERAGEX(VALUES(Dates[Mois]), [CA]) calcule la moyenne des mois réellement présents. Diviser par 3 sous-es‐

time la moyenne en début de période, quand il n'y a qu'un ou deux mois de données.

03 Pièges à éviter

⚠️ PIÈGE

Le premier mois de la série

En janvier, la fenêtre de 3 mois remonte en novembre et décembre de l'année précédente. S'ils n'existent pas dans

les données, la moyenne ne porte que sur un mois.

⚠️ PIÈGE

MAX(Ventes[Date]) au lieu de MAX(Dates[Date])

La date de départ doit venir de la table de dates. Avec MAX(Ventes[Date]), un mois sans vente fausse toute la

fenêtre.

01 Définition

EN UNE PHRASE

RANKX(table, expression, , ordre, ex aequo) calcule l'expression pour chaque ligne de la table, puis

donne la position de la valeur courante. TOPN(n, table, expression) renvoie les n lignes de la table

qui ont la plus grande valeur.

02 Le simulateur

Compare RANKX avec ALL et sans ALL.

FICHE DE COURS GRATUITE

DAX | CLASSER

RANKX et TOPN

RANKX classe, TOPN sélectionne. Le premier piège est

célèbre : sans ALL, chaque produit n'est comparé qu'à lui-

même et tout le monde finit premier. Le simulateur le

montre.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau 3 / 3, Avancé

Durée de lecture 9 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
rankx-topn.html

RANKX(ALL(...))

Rang = IF(ISINSCOPE(Produits[Produit]), RANKX(ALL(Produits[Produit]), [CA]))

MATRICE : RANG

Produit Rang

Clavier 4

Souris 5

Écran 2

Bureau 1

Chaise 3

Lampe 6

Ramette 7

Agenda 8

Total

RANKX SANS ALL (BUG)

Rang = RANKX(Produits[Produit], [CA])

MATRICE : RANG

Produit Rang

Clavier 1

Souris 1

Écran 1

Bureau 1

Chaise 1

Lampe 1

Ramette 1

Agenda 1

Total

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/rankx-topn.html

RANG ET CA DU TOP 3

Rang = IF(ISINSCOPE(Produits[Produit]), RANKX(ALL(Produits[Produit]), [CA]))

CA top 3 = CALCULATE([CA], TOPN(3, ALL(Produits[Produit]), [CA]))

Rang dense = RANKX(ALL(Produits[Produit]), [CA], , DESC, DENSE)

03 À toi de jouer

DÉFI DAX N°8 | 12 MIN

Le podium des clients

Classe les clients selon leur CA 2026, et crée une mesure qui affiche le CA des 3 meilleurs clients.

Ce que tu dois obtenir

Client CA 2026 Rang

Garage Martin 12 277 € 1

Cabinet Roy 11 590 € 2

Mairie de Lomme 9 078 € 3

Café des Arts 6 815 € 4

Librairie Page 5 118 € 5

CA top 3 32 945 €

Résultat attendu : 10 clients classés ; le total de la mesure CA top 3 vaut la somme des 3 premiers.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/rankx-topn.html

04 Pièges à éviter

⚠️ PIÈGE

Le rang 1 partout

Sans ALL, la table de comparaison est filtrée par la ligne du visuel : elle ne contient qu'un produit.

RANKX(ALL(Produits[Produit]), …) compare à tous les produits.

⚠️ PIÈGE

Un rang sur la ligne Total

Sur la ligne Total, RANKX renvoie 1, ce qui n'a aucun sens. ISINSCOPE(Produits[Produit]) permet de ne rien afficher

quand la colonne n'est pas dans le niveau en cours.

01 Définition

EN UNE PHRASE

SWITCH(TRUE(), condition1, résultat1, condition2, résultat2, …, défaut) teste les conditions dans

l'ordre et renvoie le résultat de la première vraie, ou la valeur par défaut si aucune ne l'est.

FICHE DE COURS GRATUITE

DAX | CLASSER

SWITCH(TRUE()) :
segmenter
Trois seuils, quatre résultats : avec IF, on obtient trois IF

imbriqués et une formule illisible. SWITCH(TRUE()) aligne

les conditions les unes sous les autres et renvoie le résultat

de la première vraie.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
switch-true.html

02 Le simulateur

SEGMENT CLIENT

Segment =
VAR ca = [CA]
RETURN
    SWITCH(
        TRUE(),
        ISBLANK(ca), BLANK(),
        ca >= 1500, "Premium",
        ca >= 800, "Fidèle",
        "Occasionnel"
    )

MATRICE : SEGMENT

Client Segment

Garage Martin Premium

Cabinet Roy Premium

Fleurs & Co Fidèle

Café des Arts Premium

Librairie Page Fidèle

Atelier Lumière

Total

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/switch-true.html

03 IF imbriqués ou SWITCH

LA MÊME RÈGLE, DEUX ÉCRITURES

Segment IF = IF([CA] >= 1500, "Premium", IF([CA] >= 800, "Fidèle", "Occasionnel"))

Segment =
SWITCH(
    TRUE(),
    [CA] >= 1500, "Premium",
    [CA] >= 800, "Fidèle",
    "Occasionnel"
)

04 À toi de jouer

DÉFI DAX N°9 | 10 MIN

Segmenter les clients

Classe chaque client selon son CA 2026 : Premium à partir de 6 000 €, Fidèle à partir de 2 500 €, Occasion‐

nel en dessous. Un client sans achat ne doit rien afficher.

Ce que tu dois obtenir

Client CA 2026 Segment

Garage Martin 12 277 € Premium

Cabinet Roy 11 590 € Premium

Fleurs & Co 1 408 € Occasionnel

Café des Arts 6 815 € Premium

Librairie Page 5 118 € Fidèle

Atelier Lumière 3 553 € Fidèle

…

Résultat attendu : 10 clients, 3 segments.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/switch-true.html

05 Pièges à éviter

⚠️ PIÈGE

Les seuils dans le mauvais ordre

SWITCH s'arrête à la première condition vraie. Tester « >= 800 » avant « >= 1500 » classe tous les gros clients en

Fidèle. Commence toujours par le seuil le plus haut.

⚠️ PIÈGE

Un client sans achat classé Occasionnel

Sans test ISBLANK en premier, un client sans aucune vente reçoit la valeur par défaut. Il apparaît alors dans toutes

les matrices, avec un segment qui n'a pas de sens.

01 Définition

EN UNE PHRASE

SELECTEDVALUE(colonne, remplacement) renvoie la valeur de la colonne s'il n'en reste qu'une

seule dans le contexte de filtre, et la valeur de remplacement sinon. HASONEVALUE(colonne) ren‐

voie VRAI dans ce même cas.

02 Le simulateur

TITRE DYNAMIQUE

Titre = "CA de " & SELECTEDVALUE(Produits[Catégorie], "toutes les catégories")

MATRICE : TITRE

Catégorie Titre

Informatique CA de Informatique

Mobilier CA de Mobilier

Papeterie CA de Papeterie

Total CA de toutes les catégories

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/selectedvalue.html

FICHE DE COURS GRATUITE

DAX | CLASSER

SELECTEDVALUE et
HASONEVALUE

« CA de Mobilier » quand on choisit Mobilier, « CA de toutes

les catégories » sinon : un titre qui suit les segments rend un

rapport bien plus clair. SELECTEDVALUE lit la sélection en

une fonction.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 6 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
selectedvalue.html

03 Utiliser la mesure comme titre

TITRE ET CHOIX DE MESURE PAR SEGMENT

Titre = "CA de " & SELECTEDVALUE(Produits[Catégorie], "toutes les catégories")

Valeur affichée =
SWITCH(
    SELECTEDVALUE('Choix'[Indicateur], "CA"),
    "CA", [CA],
    "Nb ventes", [Nb ventes],
    [CA]
)

04 Pièges à éviter

⚠️ PIÈGE

Plusieurs valeurs cochées

Si deux catégories sont cochées dans le segment, SELECTEDVALUE renvoie la valeur de remplacement. Pour affi‐

cher la liste, utilise CONCATENATEX(VALUES(Produits[Catégorie]), Produits[Catégorie], ", ").

⚠️ PIÈGE

Un texte dans une mesure numérique

Une mesure qui renvoie parfois un texte et parfois un nombre ne peut pas être utilisée dans un graphique. Garde

les mesures de titre séparées des mesures de calcul.

Crée la mesure Titre ci-dessous.1

Sélectionne le visuel, onglet Format > Général > Titre.2

Clique sur le bouton fx à côté du texte du titre, puis choisis la mesure Titre.3

01 Définition

EN UNE PHRASE

La ligne Total d'un visuel n'additionne pas les lignes : elle recalcule la mesure avec le contexte du to‐

tal, c'est-à-dire sans le filtre de la ligne. Dès que la mesure contient une condition ou un seuil, le total

diffère de la somme des lignes.

02 Le simulateur

Clique sur la ligne Total dans les deux versions de la mesure.

FICHE DE COURS GRATUITE

DAX | DÉPANNER

« Mon total est faux »
et les erreurs
courantes

Toutes les lignes sont justes, mais le total est faux. Ce n'est

pas un bug de Power BI : la ligne Total recalcule la mesure

dans son propre contexte, au lieu d'additionner les lignes.

Une fois qu'on l'a compris, la correction tient en une ligne.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 10 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
total-faux-erreurs.htm
l

PRIME (TOTAL FAUX)

Prime = IF([CA] > 1000, [CA] * 5%, 0)

MATRICE : PRIME

Client Prime

Garage Martin 77,05 €

Cabinet Roy 86,80 €

Fleurs & Co 56,20 €

Café des Arts 89,15 €

Librairie Page 0 €

Atelier Lumière 0 €

Total 358,65 €

PRIME (TOTAL JUSTE)

Prime =
SUMX(
    VALUES(Clients[Client]),
    IF([CA] > 1000, [CA] * 5%, 0)
)

MATRICE : PRIME

Client Prime

Garage Martin 77,05 €

Cabinet Roy 86,80 €

Fleurs & Co 56,20 €

Café des Arts 89,15 €

Librairie Page 0 €

Atelier Lumière 0 €

Total 309,20 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/total-faux-erreurs.html

03 La correction

Il faut forcer le calcul ligne par ligne, même sur le total, avec un itérateur sur la colonne des lignes du visuel :

UNE PRIME CALCULÉE CLIENT PAR CLIENT

Prime =
SUMX(
    VALUES(Clients[Client]),
    VAR ca = [CA]
    RETURN IF(ca > 1000, ca * 5%, 0)
)

04 Les autres erreurs courantes

Erreur Cause Solution

Une dépendance circulaire a été détectée Deux

colonnes

calculées

se

référencent,

souvent via

CALCULATE

Passer l'un

deux en me

ou la calcu

dans Powe

Query

La colonne … est introuvable Colonne

renommée

dans Power

Query, ou

faute de

frappe

Corriger le 

avec l'aide 

saisie semi

automatiqu

Impossible de déterminer une valeur unique pour la colonne Une

colonne

utilisée

dans une

mesure

sans

agrégation

Agréger (S

MAX) ou ut

SELECTEDV

Une table de plusieurs valeurs a été fournie là où une valeur unique était attendue VALUES ou

FILTER

renvoie

plusieurs

lignes dans

un calcul

Agréger le

résultat, ou

vérifier ave

HASONEVA

05 À toi de jouer

DÉFI DAX N°10 | 15 MIN

La commission qui ne s'additionne pas

Le commercial touche 3 % du CA 2026 d'un client quand ce CA dépasse 5 000 €. La mesure IF([CA] >

5000, [CA] * 3%, 0) donne des lignes justes, mais le total est faux. Corrige-la.

Ce que tu dois obtenir

Client CA 2026 Commission

Garage Martin 12 277 € 368,31 €

Cabinet Roy 11 590 € 347,70 €

Mairie de Lomme 9 078 € 272,34 €

Café des Arts 6 815 € 204,45 €

Total attendu 60 443 € 1 497,09 €

Résultat attendu : Avec la mesure de départ, le total affiche 1 813,29 € : il applique la règle au CA de tous les clients réunis.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/total-faux-erreurs.html

06 Pièges à éviter

⚠️ PIÈGE

Masquer le total au lieu de le corriger

Désactiver la ligne Total cache le problème mais ne le règle pas : la même mesure, utilisée dans une carte ou un

autre visuel, donnera le chiffre faux. Corrige la mesure.

⚠️ PIÈGE

Itérer sur la mauvaise colonne

SUMX(VALUES(Clients[Région]), …) calcule la prime par région, pas par client. L'itérateur doit porter sur la colonne

au niveau de laquelle la règle s'applique.

01 Définition

EN UNE PHRASE

Une mesure de couleur renvoie un texte : un code hexadécimal (« #1F8A4C ») ou un nom de couleur

(« green »). Branchée dans Mise en forme conditionnelle > Style de format : Valeur du champ, elle

colore l'arrière-plan, la police, une icône ou une barre.

02 Le simulateur

Change l'objectif : la mesure Couleur CA est recalculée et la matrice se recolore.

FICHE DE COURS GRATUITE

DAX | MISE EN FORME

Une mesure qui
renvoie une couleur
Les règles de mise en forme conditionnelle de Power BI sont

limitées à des seuils fixes. Une mesure qui renvoie un code

couleur fait tout le reste : objectifs différents par région,

comparaison à N-1, couleurs de la charte.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 8 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
mfc-couleur-par-mesur
e.html

OBJECTIF 500 €

Couleur CA =
VAR ca = [CA]
RETURN
    SWITCH(
        TRUE(),
        ISBLANK(ca), BLANK(),
        ca >= 500, "#1F8A4C",
        ca >= 300, "#E8A317",
        "#C8372D"
    )

MATRICE : CA

Catégorie Nord Sud Est Total

Informatique 714 € 603 € 98 € 1 415 €

Mobilier 610 € 874 € 1 075 € 2 559 €

Papeterie 150 € 180 € 330 €

Total 1 474 € 1 657 € 1 173 € 4 304 €

OBJECTIF 800 €

Couleur CA =
VAR ca = [CA]
RETURN
    SWITCH(
        TRUE(),
        ISBLANK(ca), BLANK(),
        ca >= 800, "#1F8A4C",
        ca >= 480, "#E8A317",
        "#C8372D"
    )

MATRICE : CA

Catégorie Nord Sud Est Total

Informatique 714 € 603 € 98 € 1 415 €

Mobilier 610 € 874 € 1 075 € 2 559 €

Papeterie 150 € 180 € 330 €

Total 1 474 € 1 657 € 1 173 € 4 304 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/mfc-couleur-par-mesure.html

03 Brancher la mesure sur le visuel

Crée la mesure de couleur, avec SWITCH(TRUE()) et des codes hexadécimaux.1

COULEUR SELON UN OBJECTIF PROPRE À CHAQUE RÉGION

Couleur objectif =
VAR ca = [CA]
VAR objectif = SELECTEDVALUE(Objectifs[Montant])
RETURN
    SWITCH(
        TRUE(),
        ISBLANK(ca) || ISBLANK(objectif), BLANK(),
        ca >= objectif, "#1F8A4C",
        ca >= objectif * 0.8, "#E8A317",
        "#C8372D"
    )

04 À toi de jouer

DÉFI DAX N°11 | 12 MIN

Colorer l'évolution

Dans une matrice par catégorie (segment Année = 2026), colore l'évolution par rapport à 2025 : vert à partir

de +30 %, orange entre 0 et +30 %, rouge en cas de baisse. La couleur doit venir d'une mesure DAX.

Ce que tu dois obtenir

Catégorie Évolution Couleur

Informatique 60,0 % Vert (#1F8A4C)

Mobilier 20,3 % Orange (#E8A317)

Papeterie -15,5 % Rouge (#C8372D)

Résultat attendu : Une mesure Évolution %, une mesure Couleur évolution, et la mise en forme conditionnelle réglée sur «

Valeur du champ ».

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/mfc-couleur-par-mesure.html

Sélectionne le visuel, onglet Format > Éléments de cellule, choisis la série à colorer.2

Active Couleur d'arrière-plan (ou Couleur de police), clique sur fx.3

Style de format : Valeur du champ, puis choisis la mesure de couleur.4

05 Pièges à éviter

⚠️ PIÈGE

Des couleurs illisibles

Un rouge vif sous un texte noir se lit mal, et un code couleur seul exclut les personnes daltoniennes. Préfère des

fonds pâles, et ajoute une icône ou un texte quand la couleur porte un sens important.

⚠️ PIÈGE

La couleur sur la ligne Total

Sur le total, l'objectif d'une seule région n'a pas de sens. Renvoie BLANK() quand plusieurs régions sont visibles

(HASONEVALUE) pour laisser le total sans couleur.

01 Définition

EN UNE PHRASE

UNICHAR(code) renvoie le caractère Unicode correspondant au code : 9650 pour ▲, 9660 pour ▼,

9658 pour ►, 9733 pour ★. Concaténé avec FORMAT, il crée un indicateur texte dans une mesure.

02 Le simulateur

FLÈCHE + ÉVOLUTION

Tendance =
VAR evo = [Évolution %]
RETURN
    SWITCH(
        TRUE(),
        ISBLANK([CA N-1]), "★ Nouveau",
        evo > 0.05, UNICHAR(9650) & " " & FORMAT(evo, "0.0 %"),
        evo < -0.05, UNICHAR(9660) & " " & FORMAT(evo, "0.0 %"),
        UNICHAR(9658) & " " & FORMAT(evo, "0.0 %")
    )

MATRICE : TENDANCE

Catégorie Tendance

Informatique ▲ 36,5 %

Mobilier ▲ 39,7 %

Papeterie ★ Nouveau

Total ▲ 50,0 %

FICHE DE COURS GRATUITE

DAX | MISE EN FORME

Icônes, flèches et
indicateurs
Un ▲ vert à côté d'un +12 % se lit en une demi-seconde.

UNICHAR insère n'importe quel symbole dans une mesure

texte, et la mise en forme conditionnelle s'occupe de la

couleur.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
icones-unichar.html

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/icones-unichar.html

03 Les codes utiles

Code Symbole Usage

UNICHAR(9650) ▲ Hausse

UNICHAR(9660) ▼ Baisse

UNICHAR(9658) ► Stable

UNICHAR(9679) ● Pastille de statut

UNICHAR(9733) ★ Nouveau, favori

UNICHAR(10004) ✔ Objectif atteint

💡 ASTUCE

Les icônes intégrées de Power BI

Mise en forme conditionnelle > Icônes : les règles permettent d'associer une icône intégrée à des seuils, sans DAX.

Avec le style Valeur du champ, une mesure peut aussi renvoyer le nom d'une icône intégrée, ou l'URL d'une image.

04 Pièges à éviter

⚠️ PIÈGE

Une mesure texte dans un graphique

Une mesure qui renvoie « ▲ 12,3 % » est un texte : elle ne peut pas servir d'axe ni de valeur dans un graphique.

Garde une mesure numérique pour les graphiques, et la mesure texte pour les tableaux et les cartes.

⚠️ PIÈGE

Le tri sur la colonne texte

Trier une matrice sur la mesure Tendance trie par ordre alphabétique des symboles. Trie sur la mesure Évolution %,

même si elle est masquée.

01 Définition

EN UNE PHRASE

REPT(texte, n) répète un texte n fois : avec le bloc plein UNICHAR(9608), on dessine une barre pro‐

portionnelle. Une mesure qui renvoie un code SVG préfixé par « data:image/svg+xml;utf8, », catégo‐

risée en URL d'image, affiche une vraie image dans la cellule.

02 Le simulateur

BARRE EN CARACTÈRES

Barre =
VAR ca = [CA]
VAR maxi = MAXX(ALL(Clients[Client]), [CA])
RETURN
    IF(NOT ISBLANK(ca),
        REPT(UNICHAR(9608), ROUND(DIVIDE(ca, maxi) * 20, 0)) & " " & FORMAT(ca, "# ##0 €"))

MATRICE : BARRE

Client Barre

Garage Martin █████████████████ 1 541 €

Cabinet Roy ███████████████████ 1 736 €

Fleurs & Co █████████████ 1 124 €

Café des Arts ████████████████████ 1 783 €

Librairie Page ███████████ 989 €

Atelier Lumière

Total

FICHE DE COURS GRATUITE

DAX | MISE EN FORME

Barres de données et
images SVG

Les barres de données intégrées de Power BI sont pratiques

mais rigides. Une mesure DAX peut dessiner sa propre barre

: en caractères pour aller vite, en SVG pour un rendu sur

mesure.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau 3 / 3, Avancé

Durée de lecture 8 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
barres-de-donnees-svg.
html

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/barres-de-donnees-svg.html

03 La version SVG

UNE BARRE SVG PROPORTIONNELLE

Barre SVG =
VAR ratio = DIVIDE([CA], MAXX(ALL(Clients[Client]), [CA]))
VAR largeur = ROUND(ratio * 100, 0)
RETURN
    "data:image/svg+xml;utf8," &
    "<svg xmlns='http://www.w3.org/2000/svg' width='100' height='14'>" &
    "<rect width='" & largeur & "' height='14' rx='3' fill='%232E52FF'/>" &
    "</svg>"

04 Pièges à éviter

⚠️ PIÈGE

Le caractère # dans le SVG

Dans une URL, # coupe le texte. Écris les couleurs avec %23 à la place de # : %232E52FF pour #2E52FF.

⚠️ PIÈGE

La référence MAXX sans ALL

Sans ALL, le maximum est calculé sur la ligne en cours seulement : toutes les barres font 100 %. Le maximum doit

porter sur tous les clients.

Crée la mesure.1

Sélectionne-la, puis Outils de mesure > Catégorie de données : URL d'image.2

Place-la dans une matrice ou un tableau, et règle la taille d'image dans Format > Valeurs.3

01 Définition

EN UNE PHRASE

Une table calculée est créée par une formule DAX qui renvoie une table (Modélisation > Nouvelle

table). Elle est calculée à l'actualisation et stockée dans le modèle, comme une table importée : elle

ne réagit pas aux segments.

02 Le simulateur

La table affichée est Produits. Clique sur une ligne du résultat pour retrouver sa ligne source.

FICHE DE COURS GRATUITE

DAX | TABLES

Les tables calculées

Une formule DAX peut renvoyer une table entière : une liste

filtrée, une table de dates, une table de paramètres.

Pratique, à condition de savoir qu'elle est figée jusqu'à la

prochaine actualisation.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
tables-calculees.html

PRODUITS CHERS

Produits chers = FILTER(Produits, Produits[Prix] >= 100)

MATRICE : MESURE

Produit Catégorie Prix

Écran Informatique 189 €

Bureau Mobilier 320 €

Chaise Mobilier 145 €

PRODUITS INFORMATIQUE

Produits Informatique = FILTER(Produits, Produits[Catégorie] = "Informatique")

MATRICE : MESURE

Produit Catégorie Prix

Clavier Informatique 49 €

Souris Informatique 25 €

Écran Informatique 189 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/tables-calculees.html

03 Les bons usages

Usage Exemple

Table de dates CALENDAR(DATE(2025,1,1), DATE(2026,12,31))

Liste de référence DISTINCT(Ventes[Code client])

Table de paramètres Voir DATATABLE et GENERATESERIES

Table de synthèse pour un export Voir SUMMARIZECOLUMNS

Table de rôle (dates de livraison) Dates livraison = Dates

Table calculée

Écrite en DAX, calculée après le chargement, à partir

des tables du modèle. Pratique pour dériver une table

d'une autre.

Power Query

Écrite en M, calculée au chargement, à partir des

sources. Mieux compressée, et la règle quand la table

peut être créée dès l'import.

04 Pièges à éviter

⚠️ PIÈGE

Attendre qu'elle suive les segments

Une table calculée ne bouge pas quand on clique sur un segment : elle est figée à l'actualisation. Pour un résultat

dynamique, il faut une mesure.

⚠️ PIÈGE

Les dépendances circulaires

Une table calculée qui utilise une mesure, elle-même basée sur cette table, crée une dépendance circulaire. Garde

des tables calculées simples, basées sur des tables importées.

01 Définition

EN UNE PHRASE

SUMMARIZECOLUMNS(colonnes de regroupement, "Nom", expression, …) renvoie une table avec

une ligne par combinaison existante des colonnes de regroupement et une colonne par expression.

ADDCOLUMNS(table, "Nom", expression, …) ajoute des colonnes calculées à une table existante.

02 Le simulateur

Clique sur une ligne de la table produite : les ventes qui la composent s'allument.

FICHE DE COURS GRATUITE

DAX | TABLES

SUMMARIZECOLUMNS

et ADDCOLUMNS

SUMMARIZECOLUMNS fait en une fonction ce qu'un tableau croisé

fait à l'écran : une ligne par combinaison, avec ses totaux.

ADDCOLUMNS ajoute des colonnes calculées à n'importe quelle

table. Ensemble, ils construisent toutes les synthèses.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau 3 / 3, Avancé

Durée de lecture 9 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
summarizecolumns-addco
lumns.html

SUMMARIZECOLUMNS

Synthèse =
SUMMARIZECOLUMNS(
    Produits[Catégorie],
    Clients[Région],
    "CA", [CA],
    "Nb ventes", COUNTROWS(Ventes)
)

MATRICE : MESURE

Combinaison Catégorie Région CA Nb ventes

Informatique | Nord Informatique Nord 1 148 € 6

Informatique | Sud Informatique Sud 1 206 € 6

Informatique | Est Informatique Est 98 € 1

Mobilier | Nord Mobilier Nord 1 367 € 4

Mobilier | Sud Mobilier Sud 1 339 € 5

Mobilier | Est Mobilier Est 1 685 € 4

Papeterie | Nord Papeterie Nord 150 € 2

Papeterie | Sud Papeterie Sud 180 € 2

ADDCOLUMNS + VALUES

Par catégorie =
ADDCOLUMNS(
    VALUES(Produits[Catégorie]),
    "CA", [CA],
    "Nb ventes", COUNTROWS(RELATEDTABLE(Ventes)),
    "Nb clients", DISTINCTCOUNT(Ventes[Code client])
)

MATRICE : MESURE

Catégorie CA Nb ventes Nb clients

Informatique 2 452 € 13 5

Mobilier 4 391 € 13 5

Papeterie 330 € 4 4

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/summarizecolumns-addcolumns.html

03 Une table virtuelle dans une mesure

Ces fonctions servent aussi à l'intérieur d'une mesure, comme table intermédiaire jamais stockée :

MOYENNE DU CA PAR CLIENT

CA moyen par client =
AVERAGEX(
    ADDCOLUMNS(VALUES(Clients[Client]), "@CA", [CA]),
    [@CA]
)

💡 ASTUCE

Préfixe les colonnes virtuelles

Nommer « @CA » une colonne ajoutée par ADDCOLUMNS la distingue au premier coup d'œil de la mesure [CA].

C'est une convention répandue, pas une obligation.

04 À toi de jouer

DÉFI DAX N°12 | 12 MIN

La table de synthèse annuelle

Crée une table calculée qui donne, pour chaque année et chaque catégorie, le CA et le nombre de ventes.

Elle servira de source à un petit tableau d'export.

Ce que tu dois obtenir

Année Catégorie CA Nb ventes

2025 Informatique 14 703 € 55

2025 Mobilier 20 581 € 59

2025 Papeterie 14 399 € 66

2026 Informatique 23 525 € 85

2026 Mobilier 24 751 € 78

2026 Papeterie 12 167 € 65

Résultat attendu : 6 lignes : 2 années × 3 catégories.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/summarizecolumns-addcolumns.html

05 Pièges à éviter

⚠️ PIÈGE

SUMMARIZE avec des colonnes calculées

L'ancienne fonction SUMMARIZE accepte des expressions, mais leurs résultats peuvent être faux dans certains

contextes. Regroupe avec SUMMARIZE si besoin, et ajoute les calculs avec ADDCOLUMNS ou

SUMMARIZECOLUMNS.

⚠️ PIÈGE

SUMMARIZECOLUMNS dans une mesure

Dans les anciennes versions de Power BI, SUMMARIZECOLUMNS ne fonctionnait pas dans une mesure soumise à

une transition de contexte. Les versions récentes le permettent, mais ADDCOLUMNS(VALUES(...)) reste le choix le

plus sûr dans une mesure.

01 Définition

EN UNE PHRASE

GENERATESERIES(début, fin, pas) crée une table d'une colonne de valeurs régulières. DATATABLE

crée une petite table écrite à la main, avec ses colonnes et ses types. Ces tables, non reliées au mo‐

dèle, alimentent un segment que les mesures lisent avec SELECTEDVALUE.

02 Le simulateur

Change la valeur du segment Remise : le CA remisé se recalcule.

FICHE DE COURS GRATUITE

DAX | TABLES

DATATABLE et
GENERATESERIES :
scénarios « et si »

« Et si on accordait 10 % de remise ? » Une petite table de

valeurs dans un segment, une mesure qui lit la valeur

choisie : le rapport devient un simulateur. Power BI le fait

avec Nouveau paramètre, et DAX le fait en une ligne.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
datatable-generateseri
es.html

REMISE 0 %

Remise = GENERATESERIES(0, 0.2, 0.05)

CA remisé = [CA] * (1 - SELECTEDVALUE(Remise[Valeur], 0))

MATRICE : CA REMISÉ

Catégorie CA remisé

Informatique 2 452 €

Mobilier 4 391 €

Papeterie 330 €

Total 7 173 €

REMISE 5 %

Remise = GENERATESERIES(0, 0.2, 0.05)

CA remisé = [CA] * (1 - SELECTEDVALUE(Remise[Valeur], 0))

MATRICE : CA REMISÉ

Catégorie CA remisé

Informatique 2 329,40 €

Mobilier 4 171,45 €

Papeterie 313,50 €

Total 6 814,35 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/datatable-generateseries.html

UNE TABLE DE SCÉNARIOS ÉCRITE À LA MAIN

Scénarios =
DATATABLE(
    "Scénario", STRING,
    "Croissance", DOUBLE,
    {
        {"Prudent", 0.02},
        {"Central", 0.05},
        {"Ambitieux", 0.10}
    }
)

CA projeté = [CA] * (1 + SELECTEDVALUE(Scénarios[Croissance], 0))

💡 ASTUCE

Modélisation > Nouveau paramètre

Le menu Nouveau paramètre > Plage numérique crée en un clic la table GENERATESERIES, la mesure qui lit la va‐

leur et le segment. Pratique pour démarrer, et la formule générée est un bon exemple à étudier.

03 Pièges à éviter

⚠️ PIÈGE

Relier la table de paramètres

Une table de paramètres ne doit pas être reliée au modèle : elle sert seulement à choisir une valeur. Une relation fil‐

trerait les ventes de façon absurde.

⚠️ PIÈGE

Oublier la valeur par défaut

Sans deuxième argument, SELECTEDVALUE renvoie vide quand rien n'est coché, et toute la mesure devient vide.

Prévois une valeur par défaut : 0 pour une remise, "Central" pour un scénario.

01 Définition

EN UNE PHRASE

INTERSECT(A, B) garde les lignes présentes dans les deux tables, EXCEPT(A, B) celles de A ab‐

sentes de B, UNION(A, B) empile les deux tables (doublons compris). Les tables doivent avoir le

même nombre de colonnes.

FICHE DE COURS GRATUITE

DAX | TABLES

UNION, EXCEPT,
INTERSECT

Qui a acheté l'an dernier et cette année ? Qui n'est pas

revenu ? Qui est nouveau ? Trois questions, trois fonctions

d'ensemble, appliquées à deux listes de clients.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau 3 / 3, Avancé

Durée de lecture 8 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
union-except-intersec
t.html

02 Le simulateur

INTERSECT : CLIENTS FIDÈLES

Fidèles =
INTERSECT(
    CALCULATETABLE(VALUES(Ventes[Code client]), Dates[Année] = 2025),
    CALCULATETABLE(VALUES(Ventes[Code client]), Dates[Année] = 2026)
)

MATRICE : MESURE

Client Années d'achat

Cabinet Roy 2025, 2026

Café des Arts 2025, 2026

Fleurs & Co 2025, 2026

Garage Martin 2025, 2026

Librairie Page 2025, 2026

EXCEPT : CLIENTS PERDUS

Perdus =
EXCEPT(
    CALCULATETABLE(VALUES(Ventes[Code client]), Dates[Année] = 2025),
    CALCULATETABLE(VALUES(Ventes[Code client]), Dates[Année] = 2026)
)

MATRICE : MESURE

Client Années d'achat

(aucun client perdu)

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/union-except-intersect.html

03 Dans une mesure : compter

Ces fonctions renvoient des tables. Dans une mesure, on compte leurs lignes avec COUNTROWS, et le résultat

suit les segments :

NOMBRE DE CLIENTS PERDUS D'UNE ANNÉE SUR L'AUTRE

Clients perdus =
VAR avant = CALCULATETABLE(VALUES(Ventes[Code client]), SAMEPERIODLASTYEAR(Dates[Date]))
VAR maintenant = VALUES(Ventes[Code client])
RETURN
    COUNTROWS(EXCEPT(avant, maintenant))

Question Fonction

Clients fidèles (N-1 et N) INTERSECT(avant, maintenant)

Clients perdus (N-1, pas N) EXCEPT(avant, maintenant)

Nouveaux clients (N, pas N-1) EXCEPT(maintenant, avant)

Tous les clients actifs DISTINCT(UNION(avant, maintenant))

04 Pièges à éviter

⚠️ PIÈGE

L'ordre des arguments d'EXCEPT

EXCEPT(A, B) et EXCEPT(B, A) ne répondent pas à la même question : l'un donne les clients perdus, l'autre les nou‐

veaux. Relis la phrase « dans A mais pas dans B ».

⚠️ PIÈGE

Les doublons d'UNION

UNION ne supprime pas les doublons, contrairement au UNION du SQL. Entoure-la de DISTINCT si tu veux une liste

unique.

01 Définition

EN UNE PHRASE

La sécurité au niveau des lignes (RLS, Row-Level Security) restreint les lignes visibles selon l'utilisa‐

teur. Un rôle porte une règle DAX sur une table, une expression vraie ou fausse pour chaque ligne :

seules les lignes où elle est vraie restent visibles, et le filtre se propage par les relations.

02 Le simulateur

Change d'utilisateur : la table masque les lignes interdites, et la matrice ne calcule que sur ce qui reste.

FICHE DE COURS GRATUITE

DAX | SÉCURITÉ

La RLS : créer un rôle

Un seul rapport, et chaque responsable régional ne voit que

sa région. La sécurité au niveau des lignes filtre le modèle

avant tout calcul : ni un segment ni une mesure ne peuvent

la contourner.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 8 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
rls-roles.html

AFFICHER EN TANT QUE : RÔLE NORD

-- Rôle « Nord », table Clients
[Région] = "Nord"

MATRICE : CA

Catégorie Nord Sud Est Total

Informatique 1 148 € 1 148 €

Mobilier 1 367 € 1 367 €

Papeterie 150 € 150 €

Total 2 665 € 2 665 €

AFFICHER EN TANT QUE : RÔLE SUD

-- Rôle « Sud », table Clients
[Région] = "Sud"

MATRICE : CA

Catégorie Nord Sud Est Total

Informatique 1 206 € 1 206 €

Mobilier 1 339 € 1 339 €

Papeterie 180 € 180 €

Total 2 725 € 2 725 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/rls-roles.html

03 Créer un rôle

💡 ASTUCE

Un rôle sans filtre pour la direction

Crée un rôle « Direction » avec la règle TRUE() : ses membres voient tout, sans être propriétaires de l'espace de

travail.

Modélisation > Gérer les rôles > Nouveau, nomme le rôle (Nord).1

Choisis la table à filtrer : de préférence une dimension (Clients), pas la table de faits.2

Écris la règle DAX : [Région] = "Nord" .3

Teste avec Modélisation > Afficher en tant que, coche le rôle.4

Après publication, dans le service Power BI : jeu de données > Sécurité, ajoute les membres de chaque rôle.5

04 Pièges à éviter

⚠️ PIÈGE

Un rôle par région

Avec 15 régions, 15 rôles à maintenir et à peupler. Dès qu'il y a plus de deux ou trois rôles, passe à la RLS dyna‐

mique : un seul rôle, une table de correspondance.

⚠️ PIÈGE

Les membres de l'espace de travail

La RLS ne s'applique pas aux administrateurs, membres et contributeurs de l'espace de travail : ils voient tout. Par‐

tage le rapport avec les lecteurs via une application ou le rôle Lecteur.

01 Définition

EN UNE PHRASE

USERPRINCIPALNAME() renvoie l'adresse de l'utilisateur connecté (son identifiant Microsoft 365).

Dans une RLS dynamique, une table Sécurité associe chaque adresse à ses régions, et un rôle

unique filtre cette table avec [E-mail] = USERPRINCIPALNAME().

02 Le simulateur

Change d'utilisateur : Inès voit le Nord, Hugo le Sud et l'Est, la direction tout.

FICHE DE COURS GRATUITE

DAX | SÉCURITÉ

La RLS dynamique

Plutôt qu'un rôle par région, une table dit qui voit quoi : une

ligne par adresse e-mail et par région. Un rôle unique filtre

cette table sur l'utilisateur connecté, et le modèle fait le

reste. Ajouter un commercial devient une simple ligne.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau 3 / 3, Avancé

Durée de lecture 9 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
rls-dynamique.html

CONNECTÉ : INÈS

-- Rôle « Commercial », table Sécurité
[E-mail] = USERPRINCIPALNAME()

MATRICE : CA

Catégorie Nord Sud Est Total

Informatique 1 148 € 1 148 €

Mobilier 1 367 € 1 367 €

Papeterie 150 € 150 €

Total 2 665 € 2 665 €

CONNECTÉ : HUGO

-- Rôle « Commercial », table Sécurité
[E-mail] = USERPRINCIPALNAME()

MATRICE : CA

Catégorie Nord Sud Est Total

Informatique 1 206 € 98 € 1 304 €

Mobilier 1 339 € 1 685 € 3 024 €

Papeterie 180 € 180 €

Total 2 725 € 1 783 € 4 508 €

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/rls-dynamique.html

03 La mise en place

Prépare la table Sécurité (E-mail, Région), une ligne par droit : une personne peut avoir plusieurs lignes.1

Charge-la et relie Sécurité[Région] à Clients[Région] : relation plusieurs à plusieurs, filtre de Sécurité vers

Clients.

2

Crée un rôle « Commercial » avec, sur la table Sécurité : [E-mail] = USERPRINCIPALNAME() .3

Teste avec Afficher en tant que > Autre utilisateur en saisissant une adresse.4

Dans le service, mets tous les lecteurs dans ce seul rôle.5

VARIANTE : LA RÈGLE DIRECTEMENT SUR CLIENTS

-- Rôle sur la table Clients, sans relation avec Sécurité
[Région] IN
    CALCULATETABLE(
        VALUES(Sécurité[Région]),
        Sécurité[E-mail] = USERPRINCIPALNAME()
    )

04 À toi de jouer

DÉFI DAX N°13 | 15 MIN

Chaque commercial ses régions

L'onglet Sécurité du classeur associe chaque adresse e-mail aux régions qu'elle peut voir. Crée un rôle dy‐

namique unique, puis vérifie ce que voit sophie@boutique.fr avec « Afficher en tant que ».

Ce que tu dois obtenir

Catégorie (vue de Sophie) CA

Informatique 18 095 €

Mobilier 24 210 €

Papeterie 13 176 €

Total 55 481 €

Résultat attendu : Sophie ne voit que le Nord et l'Est ; la direction voit tout.

Fichier, indices et solution en ligne

Le classeur « boutique » (ventes, produits, clients, dates), les trois indices progressifs et la so‐

lution commentée avec sa formule DAX sont sur la page en ligne.

mesetudesdecas.fr/dax/rls-dynamique.html

05 Pièges à éviter

⚠️ PIÈGE

Les majuscules dans les adresses

USERPRINCIPALNAME renvoie l'adresse telle qu'elle est enregistrée dans Microsoft 365. Une table qui contient «

Ines@Boutique.fr » ne correspondra pas toujours : normalise les adresses en minuscules, des deux côtés avec

LOWER.

⚠️ PIÈGE

USERNAME au lieu de USERPRINCIPALNAME

USERNAME() renvoie DOMAINE\utilisateur dans Power BI Desktop et l'adresse dans le service. USERPRINCIPAL‐

NAME() renvoie l'adresse partout : c'est elle qu'il faut utiliser.

01 Définition

EN UNE PHRASE

Afficher en tant que (Power BI Desktop) et Tester en tant que rôle (service Power BI) affichent le rap‐

port exactement comme le verrait un rôle ou un utilisateur donné. C'est la seule façon fiable de véri‐

fier une RLS avant de la confier aux lecteurs.

02 Le simulateur : ALL face à la RLS

Une mesure de part du total vue par Inès : même ALL ne voit que le Nord.

INÈS (RÔLE NORD) : % DU TOTAL

% du total = DIVIDE([CA], CALCULATE([CA], ALL(Ventes)))

MATRICE : % DU TOTAL

Catégorie Nord Sud Est Total

Informatique 43,1 % 43,1 %

Mobilier 51,3 % 51,3 %

Papeterie 5,6 % 5,6 %

Total 100,0 % 100,0 %

Simulateur interactif

En ligne, cette fiche contient un simulateur Power BI : on clique sur une case de la matrice et on

voit les lignes de la table qui participent au calcul, avec le contexte de filtre de la case. Scannez

le code ou rendez-vous sur la page.

mesetudesdecas.fr/dax/rls-tester-publier.html

FICHE DE COURS GRATUITE

DAX | SÉCURITÉ

Tester et publier la
RLS

Une RLS mal testée, c'est soit un commercial qui voit les

chiffres des autres, soit un directeur qui ne voit rien. Trois

vérifications et deux règles de partage suffisent à l'éviter.

Version V1.0

Dernière mise à

jour
02/11/2026

Niveau 3 / 3, Avancé

Durée de lecture 8 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/dax/
rls-tester-publier.htm
l

03 La liste de vérification

04 Pièges à éviter

⚠️ PIÈGE

Les relations bidirectionnelles

Une relation bidirectionnelle peut propager le filtre de sécurité dans un sens inattendu. Si tu en as besoin, coche «

Appliquer le filtre de sécurité dans les deux directions » en connaissance de cause, et teste chaque rôle.

⚠️ PIÈGE

Croire que masquer une page protège les données

Masquer une page ou un visuel n'est pas de la sécurité : les données restent accessibles par l'export ou Analyser

dans Excel. Seule la RLS filtre réellement les lignes.

Chaque rôle dans Afficher en tant que : les totaux correspondent-ils au périmètre attendu ?1

Un utilisateur sans droit : il doit voir un rapport vide, pas tout le rapport.2

Les mesures avec ALL : leurs « 100 % » sont-ils compréhensibles pour l'utilisateur ?3

Dans le service : jeu de données > Sécurité, membres attribués, puis Tester en tant que rôle.4

Le partage : les lecteurs passent par une application ou le rôle Lecteur, jamais Membre.5

